

Незнайка и Коллатц. 7-10 класс

Незнайка очень полюбил функцию Коллатца $f(n) = \begin{cases} \frac{n}{2}, & n - \text{четное число} \\ 3n + 1, & n - \text{нечетное число} \end{cases}$, про которую

есть недоказанная гипотеза о том, что её применение снова и снова, стартуя с любого натурального ненулевого числа, всегда приводит к числу 1, после чего возникает цикл $1 \rightarrow 4 \rightarrow 2 \rightarrow 1$. Незнайке пришло в голову, что можно применять обратное преобразование, и от какого-то натурального ненулевого числа a переходить к одному из чисел b , такому что $f(b) = a$. Он обозначил такое обратное преобразование f^{-1} и стал искать способы перейти от стартового числа n_1 к финишному числу n_2 за минимально возможное количество шагов, применяя на каждом шаге либо функцию Коллатца f , либо обратное преобразование f^{-1} к текущему результату, полученному на предыдущем шаге (на первом шаге за текущий результат Незнайка брал число n_1). Например, от $n_1 = 10$ к $n_2 = 21$ можно перейти за 5 шагов: $10 \rightarrow f(10) = 5 \rightarrow f(5) = 16 \rightarrow f^{-1}(16) = 32 \rightarrow f^{-1}(32) = 64 \rightarrow f^{-1}(64) = 21$

Полагая, что гипотеза Коллатца верна, помогите Незнайке написать программу, которая определяет минимально возможное количество шагов, за которые из данного стартового числа n_1 можно получить финишное число n_2 .

Формат ввода: В первой строке вводится натуральное число n_1 : $1 \leq n_1 \leq 5000$. Во второй строке содержится натуральное число n_2 : $1 \leq n_2 \leq 5000$.

Формат вывода: В единственной строке выводится либо наименьшее из возможных количество последовательных применений функции $f(x)$ или преобразования $f^{-1}(x)$, после которых стартовое число n_1 преобразуется в финишное число n_2 , либо -1, если стартовое число n_1 не может быть преобразовано таким способом в финишное число n_2 . Если $n_1 = n_2$, то выводится 0.

Ввод примера №1:	Ввод примера №2:	Ввод примера №3:	Ввод примера №4:
10	8	5000	2
21	1	5000	1
Вывод примера №1:	Вывод примера №2:	Вывод примера №3:	Вывод примера №4:
5	2	0	1

Решение

Ставя каждому натуральному числу в соответствие вершину графа, и соединяя дугами все пары вершин с числами a и b , такими что $f(a) = b$, мы получаем, так называемый, граф Коллатца. Из любой вершины этого графа можно двигаться только в сторону 1, путь до которой существует, если гипотеза Коллатца верна. Известно, что в диапазоне рассматриваемых чисел длина такого пути не превосходит 237. В этом можно убедиться с помощью предварительных расчётов. Длину 237 имеет путь от 3711 к 1. Находим путь к 1, ведущий из n_1 . Находим путь к 1, ведущий из n_2 . Находим вершину, находящуюся дальше всего от 1 и лежащую на обоих путях. Часть пути от n_1 до этой вершины, к которой в приписана задом наперёд часть пути от n_2 до этой вершины, соответствует искомым шагам преобразований. Сама вершина, которую мы нашли, входит туда не более чем один раз. Её не следует включать, если она совпадает с n_1 и/или с n_2 . Количество найденных шагов является ответом, который надо вывести. При поиске пути к 1 следует иметь в виду, что $4 \rightarrow f^{-1}(4) = 1$ и это короче, чем $4 \rightarrow f(4) = 2 \rightarrow f(2) = 1$. Другими словами, если n_1 и/или n_2 находятся в диапазоне от 1 до 2, то такие случаи надо анализировать отдельно. Также следует иметь в виду, что ради эффективности перед поиском путей можно проверить тривиальные случаи, когда n_1 и n_2 совпадают или находятся в одном шаге друг от друга. Составляя программу, стоит иметь в виду, что промежуточные числа могут быть довольно велики. Так, на пути от 4591 к 1 лежит число 8153620.

Код возможного решения

```
program COLLATZ710(input, output);
const   PATH_LENGTH = 238;
type    massiv = array [1..PATH_LENGTH] of cardinal;
        path = record length : 0..PATH_LENGTH; numbers : massiv end;
function F(X : cardinal) : cardinal;
begin
    if (X mod 2) = 0 then F := X div 2 else F := 3 * X + 1
end;
```

```

procedure APPEND (V : cardinal; var P : PATH);
begin with P do begin
    length := length + 1;
    numbers[length] := V
end
end;
procedure PATHTO1 (N : cardinal; var P : path);
begin
    P.length := 0;
    while N > 1 do begin
        APPEND(N, P);
        N := F(N);
    end;
    APPEND(1, P)
end;

var N1, N2 : cardinal;
    I, J, RESULT : cardinal;
    P1, P2 : path;
begin
    readln(N1);
    read(N2);
    if N1=N2 then write(0) else begin
        if N1 > 1 then I := F(N1) else I := N1;
        if N2 > 1 then J := F(N2) else J := N2;
        if (I = N2) or (J = N1) then write(1) else begin
            PATHTO1(N1, P1);
            PATHTO1(N2, P2);
            I := P1.length;
            J := P2.length;
            while (I > 0) and (J > 0) and (P1.numbers[I] = P2.numbers[J]) do begin
                I := I - 1;
                J := J - 1;
            end;
            RESULT := I+J;
            if ((N2 = 1) and (N1 > 2)) or ((N1 = 1) and (N2 > 2)) then RESULT := RESULT - 1;
            write(RESULT);
        end
    end
end
end.

```